Section 4.2 Projections - Examples

```
> restart: with(LinearAlgebra): with(plottools): with(plots):
```

Example 1 (Projection onto a line): Project the vector **b** = (-1,2,0) onto **a** = (1,1,2).

First, let's see these two vectors.

```
> b := <-1, 2, 0>; a := <1, 1, 2>;
> zerovec := <0,0,0>:
  plotb := line(convert(zerovec,list), convert(b,list), linestyle=1,
      thickness=2, color=red):
  labelb := textplot3d([b[1]/1.5,b[2]/1.5,b[3]/1.5, ` b`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=red):
  plota := line(convert(zerovec,list), convert(a,list), linestyle=1,
      thickness=2, color=green):
  labela := textplot3d([a[1]/1.5,a[2]/1.5,a[3]/1.5, ` a`],
      align=RIGHT, font = [HELVETICA, BOLD, 12],color=green):
  display3d({plotb, labelb, plota, labela},axes=normal,scaling=
  constrained);
```

$$b := \begin{bmatrix} -1 \\ 2 \\ 0 \end{bmatrix}$$

$$a := \begin{bmatrix} 1 \\ 1 \\ 2 \end{bmatrix}$$



Now we compute xhat, the projection **p** of **b** onto **a**, and the error vector **e** = **b** - **p**.

```
> xhat := (a.b)/(a.a);
```

$$xhat := \frac{1}{6} \tag{1}$$

```
> p := xhat * a;
```

$$p := \begin{bmatrix} \frac{1}{6} \\ \frac{1}{6} \\ \frac{1}{3} \end{bmatrix} \tag{2}$$

```
> e := b - p;
```

(3)

$$e := \begin{bmatrix} -\frac{7}{6} \\ \frac{11}{6} \\ -\frac{1}{3} \end{bmatrix} \quad (3)$$

Add the projection and the error vector to the plot.

```
> plotp := line(convert(zerovec,list), convert(p,list), linestyle=1,
      thickness=3, color=blue):
  labelp := textplot3d([p[1]/1.5,p[2]/1.5,p[3]/1.5, ` p`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=blue):
  plote := line(convert(p,list), convert(b,list), linestyle=1,
      thickness=2, color=brown):
  labele := textplot3d([p[1]+e[1]/2,p[2]+e[2]/2,p[3]+e[3]/2, ` e`],

      align=RIGHT, font = [HELVETICA, BOLD, 12],color=brown):
  display3d({plotb, labelb, plota, labela,
      plotp, labelp, plote,labele},axes=normal,scaling=constrained);
```




```
> aM := ~Matrix(a);
  P := Multiply(aM,Transpose(aM))/(a.a);
```

$$aM := \begin{bmatrix} 1 \\ 1 \\ 2 \end{bmatrix}$$

$$P := \begin{bmatrix} \frac{1}{6} & \frac{1}{6} & \frac{1}{3} \\ \frac{1}{6} & \frac{1}{6} & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{3} & \frac{2}{3} \end{bmatrix} \quad \textbf{(4)}$$

Check that this works for the projection of **b** onto **a**.

```
> Multiply(P,~Matrix(b));
```

$$\begin{bmatrix} \frac{1}{6} \\ \frac{1}{6} \\ \frac{1}{3} \end{bmatrix} \quad \textbf{(5)}$$

Use P to project another vector onto **a**.

```
> b2 := <4,3,2>;
  p2 := ~Vector(Multiply(P,~Matrix(<4,3,2>)));
  e2 := b2 - p2;
```

$$b2 := \begin{bmatrix} 4 \\ 3 \\ 2 \end{bmatrix}$$

$$p2 := \begin{bmatrix} \frac{11}{6} \\ \frac{11}{6} \\ \frac{11}{3} \end{bmatrix}$$

$$e2 := \begin{bmatrix} \frac{13}{6} \\ \frac{7}{6} \\ -\frac{5}{3} \end{bmatrix} \quad \textbf{(6)}$$

Visualize this projection as well.

```
> plotb2 := line(convert(zerovec,list), convert(b2,list), linestyle=
  1,
      thickness=2, color=maroon):
  labelb2 := textplot3d([b2[1]/1.5,b2[2]/1.5,b2[3]/1.5, ` b2`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=maroon):
  plotp2 := line(convert(zerovec,list), convert(p2,list), linestyle=
  1,
      thickness=1, color=violet):
  labelp2 := textplot3d([p2[1]/1.5,p2[2]/1.5,p2[3]/1.5, ` p2`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=violet):
  plote2 := line(convert(p2,list), convert(b2,list), linestyle=1,
      thickness=2, color=brown):
  labele2 := textplot3d([p2[1]+e2[1]/2,p2[2]+e2[2]/2,p2[3]+e2[3]/2, `
  e2`],
      align=RIGHT, font = [HELVETICA, BOLD, 12],color=brown):
  display3d({plotb, labelb, plota, labela,
      plotp, labelp, plote, labele, plotb2, labelb2, plotp2, labelp2,
  plote2,labele2},axes=normal,scaling=constrained);
```



Example 2 (Projection onto a plane): Project the vector **b** = (-2,1,1) onto the plane x + 2y + z = 0.

First, let's see the vector and the plane.

```
> b3 := <-2, 1, 1>; plane := x+2*y+z=0;
```

$$b3 := \begin{bmatrix} -2 \\ 1 \\ 1 \end{bmatrix}$$

$$plane := x + 2\,y + z = 0 \quad \textbf{(7)}$$

```
> plotb3 := line(convert(zerovec,list), convert(b3,list), linestyle=
1,
    thickness=4, color=red):
labelb3 := textplot3d([b3[1]/1.5,b3[2]/1.5,b3[3]/1.5, ` b3`],
    align=RIGHT, font = [HELVETICA, BOLD, 12], color=red):
plotplane := implicitplot3d(plane, x=-3..3,y=-3..3,z=-3..3, color=
green, transparency=.8):
display3d({plotb3, labelb3, plotplane},axes=normal,scaling=
constrained);
```



Next find the matrix A and compute the vector **xhat**.

```
> s2 := <subs({y=1,z=0},solve(plane,x)),1,0>;
```

$$s2 := \begin{bmatrix} -2 \\ 1 \\ 0 \end{bmatrix} \quad \textbf{(8)}$$

```
> s3 := <subs({y=0,z=1},solve(plane,x)),0,1>;
```

$$s3 := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} \tag{9}$$

```
> A := <s2|s3>;
```

$$A := \begin{bmatrix} -2 & -1 \\ 1 & 0 \\ 0 & 1 \end{bmatrix} \tag{10}$$

```
> AT := Transpose(A);
  ATb3 := Multiply(AT,b3);
```

$$AT := \begin{bmatrix} -2 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix}$$

$$ATb3 := \begin{bmatrix} 5 \\ 3 \end{bmatrix} \tag{11}$$

```
> ATAInv := MatrixInverse(Multiply(AT,A));
  xhatvec := Multiply(ATAInv,ATb3);
```

$$ATAInv := \begin{bmatrix} \frac{1}{3} & -\frac{1}{3} \\ -\frac{1}{3} & \frac{5}{6} \end{bmatrix}$$

$$xhatvec := \begin{bmatrix} \frac{2}{3} \\ \frac{5}{6} \end{bmatrix} \tag{12}$$

Then the projection is computed as A times **xhat.**

```
> p3 := Multiply(A,xhatvec);
```

$$p3 := \begin{bmatrix} -\frac{13}{6} \\ \frac{2}{3} \\ \frac{5}{6} \end{bmatrix} \tag{13}$$

```
> e3 := b3 - p3;
```

(14)

$$e3 := \begin{bmatrix} \frac{1}{6} \\ \frac{1}{3} \\ \frac{1}{6} \end{bmatrix} \quad \textbf{(14)}$$

Plot the projection and error vectors

```
> plotb3 := line(convert(zerovec,list), convert(b3,list), linestyle=
  1,
      thickness=4, color=red):
  labelb3 := textplot3d([b3[1]/1.5,b3[2]/1.5,b3[3]/1.5, ` b3`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=red):
  plotp3 := line(convert(zerovec,list), convert(p3,list), linestyle=
  1,
      thickness=4, color=blue):
  labelp3 := textplot3d([p3[1]/1.5,p3[2]/1.5,p3[3]/1.5, ` p3`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=blue):
  plote3 := line(convert(p3,list), convert(b3,list), linestyle=1,
      thickness=4, color=brown):
  labele3 := textplot3d([p3[1]+e3[1]/2,p3[2]+e3[2]/2,p3[3]+e3[3]/2, `
  e3`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=brown):
  plotplane := implicitplot3d(plane, x=-3..3,y=-3..3,z=-3..3, color=
  green, transparency=.8):
  display3d({plotb3, labelb3, plotplane, plotp3, labelp3, plote3,
  labele3},axes=normal,scaling=constrained);
```



Compute the projection matrix P and use it to project the vector b4 = (2,3,-1) onto the plane.

```
> P2 := Multiply(A, Multiply(ATAInv,AT));
```

$$P2 := \begin{bmatrix} \frac{5}{6} & -\frac{1}{3} & -\frac{1}{6} \\ -\frac{1}{3} & \frac{1}{3} & -\frac{1}{3} \\ -\frac{1}{6} & -\frac{1}{3} & \frac{5}{6} \end{bmatrix} \quad \textbf{(15)}$$

```
> b4 := <2,3,-1>;
```

$$b4 := \begin{bmatrix} 2 \\ 3 \\ -1 \end{bmatrix} \quad \textbf{(16)}$$

```
> p4 := Multiply(P2,b4);
```

(17)

$$p4 := \begin{bmatrix} \frac{5}{6} \\ \frac{2}{3} \\ -\frac{13}{6} \end{bmatrix} \quad (17)$$

```
> e4 := b4 - p4;
```

$$e4 := \begin{bmatrix} \frac{7}{6} \\ \frac{7}{3} \\ \frac{7}{6} \end{bmatrix} \quad (18)$$

Add these new vectors to the plot:

```
> plotb4 := line(convert(zerovec,list), convert(b4,list), linestyle=
  1,
      thickness=4, color=maroon):
  labelb4 := textplot3d([b4[1]/1.5,b4[2]/1.5,b4[3]/1.5, ` b4`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=maroon):
  plotp4 := line(convert(zerovec,list), convert(p4,list), linestyle=
  1,
      thickness=4, color=violet):
  labelp4 := textplot3d([p4[1]/1.5,p4[2]/1.5,p4[3]/1.5, ` p4`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=violet):
  plote4 := line(convert(p4,list), convert(b4,list), linestyle=1,
      thickness=4, color=brown):
  labele4 := textplot3d([p4[1]+e4[1]/2,p4[2]+e4[2]/2,p4[3]+e4[3]/2, `
  e4`],
      align=RIGHT, font = [HELVETICA, BOLD, 12], color=brown):
  plotplane := implicitplot3d(plane, x=-3..3,y=-3..3,z=-3..3, color=
  green, transparency=.8):
  display3d({plotb3, labelb3, plotplane, plotp3, labelp3, plote3,
  labele3, plotb4, labelb4, plotp4, labelp4, plote4, labele4},axes=
  normal,scaling=constrained);
```


3
2
1
0
−1
−2
−3
x
y
e3
p3
e4
p4